
1. [bookmark: _Toc122434485][bookmark: _Toc61465330]Overview
This document lists all the changes needed to correct issues in the ATS iWD-TTCN3-B2020-09_D20wk50 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc61465331]Corrections required

1. [bookmark: _Toc30685521]Change 1
	Function name
	f_TC_7_1_1_2_3_NR_TestBody()

	Reason for change
	1. Due to race conditions in the SS, the indication of PRACH preamble and msg3 reception can come to the TTCN ports in any order.

2. Currently UL grant is configured too late to receive RRCReconfigurationComplete at step 21

	Summary of change
	1. Add control variables, interleave and repeat statements to allow SYSTEM_IND (for RACH preamble or HARQ ACK) and DRB data indication to be received in either order at steps 6-8, 11-13,16-18, and 20-21. Check timing information of recieved ASPs to confirm correct sequencing.

2. Default UL grant allocation can be configured at the beginning of the test sequence.


	TTCN module
	MAC_NR5GC

	MCC160 Comment
	



Before Change
	function f_TC_7_1_1_2_3_NR_TestBody() runs on NR5GC_PTC
{

var NG_NAS_MSG_Indication_Type v_NasInd;
timer t_Watchdog := 5.0;


t_Watchdog.start;
// @siclog "Step 2" siclog@
SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?));
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
// @siclog "Step 3" siclog@
//Preconfigured at SS
// @siclog "Step 4-5" siclog@
alt {
[] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?))
{
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
}
[] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, (ack, nack)))
{
t_Watchdog.stop;
f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4: UE is sending HARQ ACK/NACK ");
}
};
// @siclog "Step 6" siclog@
//Preconfigured at SS
// @siclog "Step 7" siclog@
SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest));
// @siclog "Step 9" siclog@
//Preconfigured at SS
// @siclog "Step 10-11" siclog@
alt {
[] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?))
{
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 11");
}
[] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, (ack, nack)))
{
t_Watchdog.stop;
f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10: UE is sending HARQ ACK/NACK ");
}
};
// @siclog "Step 12" siclog@
//Preconfigured at SS
// @siclog "Step 13" siclog@
SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest));
// @siclog "Step 14" siclog@
//Preconfigured at SS
// @siclog "Step 15-16" siclog@
alt {
[] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?))
{
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");
}
[] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, (ack, nack)))
{
t_Watchdog.stop;
f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15: UE is sending HARQ ACK/NACK ");
}
};
// @siclog "Step 17" siclog@
//Preconfigured at SS
// @siclog "Step 18" siclog@
SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest));
// @siclog "Step 19" siclog@
//Preconfigured at SS
// @siclog "Step 20-21" siclog@
// @sic R5s200080 R5s201287 sic@
SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, ack));
// disable HARQ reception indication
f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(disable));

f_NR_ULGrantConfiguration_Start(nr_Cell1);

v_NasInd := f_NR_RRCSetupComplete_Def(nr_Cell1, cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def));

f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 21");

// disable PRACH reception indication
f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable));
f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(disable));
// @siclog "Step 22-25" siclog@
f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_NasInd);

}; 



After Change
	Import 
import from NAS_CommonTemplates all;
import from NG_NAS_Templates all; 
import from NR_ASP_SrbDefs all;
......
function f_TC_7_1_1_2_3_NR_TestBody() runs on NR5GC_PTC
{

var NG_NAS_MSG_Indication_Type v_NasInd;
timer t_Watchdog := 5.0;
var boolean v_IsPrach := false;
var boolean v_Ismsg3 := false;
var NR_SRB_COMMON_IND v_ReceivedAsp;
var NR_SRB_COMMON_IND v_SRB_COMMON_IND;
var NR_SYSTEM_IND v_SYSTEM_IND;
var SubFrameTiming_Type v_Timing_SysInd;
var SubFrameTiming_Type v_Timing_SrbInd;

t_Watchdog.start;
//Preconfigured at SS
f_NR_ULGrantConfiguration_Start(nr_Cell1);
// @siclog "Step 2" siclog@
SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?));
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
// @siclog "Step 3" siclog@
//Preconfigured at SS
// @siclog "Step 4-5" siclog@
alt {
[not v_IsPrach] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?))-> value v_SYSTEM_IND
{
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
v_IsPrach := true;
if(not v_Ismsg3)
{
repeat;
}
}
[] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, (ack, nack)))
{
t_Watchdog.stop;
f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4: UE is sending HARQ ACK/NACK ");
}
// @siclog "Step 6" siclog@
//Preconfigured at SS
// @siclog "Step 7" siclog@
[not v_Ismsg3] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest))-> value v_SRB_COMMON_IND
{
v_Ismsg3 := true;
if(not v_IsPrach)
{
repeat;
} 
}
};
v_Timing_SrbInd := valueof(v_SRB_COMMON_IND.Common.TimingInfo.SubFrame)
v_Timing_SysInd := valueof(v_SYSTEM_IND.Common.TimingInfo.SubFrame);
fl_NR_RacingCond_CheckSequence(nr_Cell1, SysIndBeforeRrcMsg, v_Timing_SrbInd, v_Timing_SysInd);
// @siclog "Step 9" siclog@
//Preconfigured at SS
// @siclog "Step 10-11" siclog@
alt {
[v_IsPrach] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?))-> value v_SYSTEM_IND 
{
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 11");
v_IsPrach := false;
if(v_Ismsg3)
{
repeat;
}
}
[] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, (ack, nack)))
{
t_Watchdog.stop;
f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10: UE is sending HARQ ACK/NACK ");
}
// @siclog "Step 12" siclog@
//Preconfigured at SS
// @siclog "Step 13" siclog@
[v_Ismsg3] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest))-> value v_SRB_COMMON_IND
{
v_Ismsg3 := false;
if(v_IsPrach)
{
repeat;
}
}
};
v_Timing_SrbInd := valueof(v_SRB_COMMON_IND.Common.TimingInfo.SubFrame);
v_Timing_SysInd := valueof(v_SYSTEM_IND.Common.TimingInfo.SubFrame);
fl_NR_RacingCond_CheckSequence(nr_Cell1, SysIndBeforeRrcMsg, v_Timing_SrbInd, v_Timing_SysInd);
// @siclog "Step 14" siclog@
//Preconfigured at SS
// @siclog "Step 15-16" siclog@
alt {
[not v_IsPrach] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?))-> value v_SYSTEM_IND
{
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");
v_IsPrach := true;
if(not v_Ismsg3)
{
repeat;
}
}
// @siclog "Step 19" siclog@
//Preconfigured at SS
// @siclog "Step 20-21" siclog@
// @sic R5s200080 R5s201287 sic@
[] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, (ack, nack)))
{
t_Watchdog.stop;
f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15: UE is sending HARQ ACK/NACK ");
}
// @siclog "Step 17" siclog@
//Preconfigured at SS
// @siclog "Step 18" siclog@
[not v_Ismsg3] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest))-> value v_SRB_COMMON_IND
{
v_Ismsg3 := true;
if(not v_IsPrach)
{
repeat;
} 
}
};
v_Timing_SrbInd := valueof(v_SRB_COMMON_IND.Common.TimingInfo.SubFrame);
v_Timing_SysInd := valueof(v_SYSTEM_IND.Common.TimingInfo.SubFrame);
fl_NR_RacingCond_CheckSequence(nr_Cell1, SysIndBeforeRrcMsg, v_Timing_SrbInd, v_Timing_SysInd);
interleave {
[] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, ack))-> value v_SYSTEM_IND
{
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 20-21 Receive Ack");
// disable HARQ reception indication
f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(disable));
//MOVED f_NR_ULGrantConfiguration_Start(nr_Cell1);

}
[] SRB.receive(car_NR_SRB1_RrcNasPdu_IND(nr_Cell1, cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def), cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected), omit)) -> value v_ReceivedAsp
{
v_NasInd := v_ReceivedAsp.Signalling.Nas[0];
f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 20-21 Receive RRCSetupComplete");
}
};
v_Timing_SrbInd := valueof(v_ReceivedAsp.Common.TimingInfo.SubFrame)
v_Timing_SysInd := valueof(v_SYSTEM_IND.Common.TimingInfo.SubFrame);
fl_NR_RacingCond_CheckSequence(nr_Cell1, SysIndBeforeRrcMsg, v_Timing_SrbInd, v_Timing_SysInd);

f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 21");

// disable PRACH reception indication
f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable));
f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(disable));//@sic R5s201287 sic@
// @siclog "Step 22-25" siclog@
f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_NasInd);

};




1. Change 2
	Function name
	f_TC_7_1_1_2_4_NR_TestBody(...)

	Reason for change
	Due to race conditions in the SS, the indication of PRACH preamble and msg3 reception can come to the TTCN ports in any order.


	Summary of change
	Add control variables, interleave and repeat statements to allow SYSTEM_IND (for RACH preamble or HARQ ACK) and DRB data indication to be received in either order at steps 9-11. Check timing information of recieved ASPs to confirm correct sequencing.


	TTCN module
	MAC_NR5GC

	MCC160 Comment
	



Before Change
	function f_TC_7_1_1_2_4_NR_TestBody(DRB_Identity p_NR_DRB_Id) runs on NR5GC_PTC
{
var NR_PhysicalParameters_Type v_NR_PhysicalParameters ;
var SubFrameTiming_Type v_SysinfoActivation;
var SubFrameTiming_Type v_TimingDL ;
var NR_Prach_RootSequenceIndex_Type v_Prach_RootSequenceIndex;
var NR_DuplexMode_Type v_NR_DuplexMode := f_NR_CellInfo_GetDuplexMode(nr_Cell1);
timer t_Watchdog := 10.0;

if (ischosen(v_NR_DuplexMode.TDD)) //FR1 TDD
{
v_Prach_RootSequenceIndex := {l139 := 2};
}
else //FR1 FDD
{
v_Prach_RootSequenceIndex := {l139 := 20};
}
// @siclog "Step 6" siclog@ preconfig SS

// @siclog "Step 7" siclog@
//@sic R5s191176 sic@
DRB.send(cas_NR_DRB_COMMON_REQ_PDCP_SDUList(nr_Cell1,
p_NR_DRB_Id,
-,
{crs_NR_PDCP_SDU_7B} ));
// @siclog "Step 8" siclog@
SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, ack));
f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(disable));
// @siclog "Step 9" siclog@
SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, -, (0..31)));
//@siclog "Step 10 and 12" siclog@
// proconfigured at SS

// @siclog "Step 11" siclog@
DRB.receive (car_NR_DRB_COMMON_IND_PDCP_SDUList(nr_Cell1, p_NR_DRB_Id, -, {crs_NR_PDCP_SDU_7B}))

};



After Change
	import from NR_ASP_DrbDefs all;
......
function f_TC_7_1_1_2_4_NR_TestBody(DRB_Identity p_NR_DRB_Id) runs on NR5GC_PTC
{
var NR_PhysicalParameters_Type v_NR_PhysicalParameters ;
var SubFrameTiming_Type v_SysinfoActivation;
var SubFrameTiming_Type v_TimingDL ;
var NR_Prach_RootSequenceIndex_Type v_Prach_RootSequenceIndex;
var NR_DuplexMode_Type v_NR_DuplexMode := f_NR_CellInfo_GetDuplexMode(nr_Cell1);
timer t_Watchdog := 10.0;
var NR_DRB_COMMON_IND v_DRB_COMMON_IND;
var NR_SYSTEM_IND v_SYSTEM_IND;
var SubFrameTiming_Type v_Timing_SysInd;
var SubFrameTiming_Type v_Timing_DrbInd;

if (ischosen(v_NR_DuplexMode.TDD)) //FR1 TDD
{
v_Prach_RootSequenceIndex := {l139 := 2};
}
else //FR1 FDD
{
v_Prach_RootSequenceIndex := {l139 := 20};
}
// @siclog "Step 6" siclog@ preconfig SS

// @siclog "Step 7" siclog@
//@sic R5s191176 sic@
DRB.send(cas_NR_DRB_COMMON_REQ_PDCP_SDUList(nr_Cell1,
p_NR_DRB_Id,
-,
{crs_NR_PDCP_SDU_7B} ));
// @siclog "Step 8" siclog@
SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, ack));
f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(disable));

interleave {
// @siclog "Step 9" siclog@
[] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, -, (0..31)))-> value v_SYSTEM_IND
{}
//@siclog "Step 10 and 12" siclog@
// proconfigured at SS
// @siclog "Step 11" siclog@
[] DRB.receive (car_NR_DRB_COMMON_IND_PDCP_SDUList(nr_Cell1, p_NR_DRB_Id, -, {crs_NR_PDCP_SDU_7B}))-> value v_DRB_COMMON_IND
{}
};
v_Timing_DrbInd := valueof(v_DRB_COMMON_IND.Common.TimingInfo.SubFrame);
v_Timing_SysInd := valueof(v_SYSTEM_IND.Common.TimingInfo.SubFrame);
fl_NR_RacingCond_CheckSequence(nr_Cell1, SysIndBeforeRrcMsg, v_Timing_DrbInd, v_Timing_SysInd);

};




