
	Reason for change:
	24.501 clause 9.11.4.8 defines the EPS Parameters List, included in the Mapped EPS Bearer Contexts IE to contain N parameters, each consisting of:
- an EPS parameter identifier, 
- the length of the EPS parameter contents and 
- the EPS parameter contents itself.

Then when the parameter identifier indicates mapped EPS QoS parameters, the length and parameter contents field are coded as specified in subclause 9.9.4.3 of 3GPP TS 24.301

However the TTCN includes 2 length fields: for the length of the EPS parameter contents AND the length of the EPS QoS contents

	
	

	Summary of change:
	Don’t include the Length of EPS parameter contents field if the parameter is a mapped EPS QoS parameter, as this will be included in the EPS QoS template.

	
	

	Consequences if not approved:
	A conformant UE may fail to decode the mapped EPS QoS parameters

	Change 1:
	

	Function name
	f_Get_NG_PDUSessionEstablishmentAccept

	Reason for change
	Two length fields are included in the MappedEPSBearerContext when the parameter is a mapped EPS QoS.

	Summary of change
	Don’t include the Length of EPS parameter contents field if the parameter is a mapped EPS QoS parameter, as this will be included in the EPS QoS template.

	TTCN module
	NG_NASTemplateFunctions



	  function f_Get_NG_PDUSessionEstablishmentAccept(GSM_MobilityInfo_Type p_GSM_MobilityInfo,
                                                  GMM_MobilityInfo_Type p_GMM_MobilityInfo,
                                                  template (value) DNN p_DNN,       // @sic R5-205939 sic@
                                                  template (value) SSC_Mode p_SSC_Mode := crs_SSC_Mode(-, '001'B),
                                                  template (value) Session_AMBR p_Session_AMBR := cs_Session_AMBR ('06'O, '05'O, '0004'O, '05'O, '0004'O),
                                                  template (omit) PDU_SessionType p_PDU_SessionType := omit,
                                                  template (omit) GMM_GSM_Cause p_Cause := omit,
                                                  template (omit) GPRS_Timer p_RQTimer := omit,
                                                  template (omit) S_NSSAI_Type p_S_NSSAI := cs_S_NSSAI_SST1eMBB_WithIEI, // @sic R5s190919 sic@
                                                  template (omit) AlwaysOnPDUSessionInd p_AlwaysOnPDUSessionInd := omit,
                                                  template (omit) EAP_Message p_EAP := omit,
                                                  template (omit) QoS_Rules p_QoS_Rules := omit,
                                                  template (omit) QoSFlowDescr p_QoSFlowDescr := omit,
                                                  template (omit) ExtdProtocolConfigOptions p_ExtdPCO := omit,
                                                  template (omit) GPRS_Timer3 p_BackOff := omit,
                                                  template (omit) charstring p_BuildQosFlow := omit,
                                                  boolean p_InterworkWithEPS := true,
                                                  template (omit) NGSM_NetworkFeatureSupport p_NGSM_NtwkFeatSupport := omit, // @sic R5s201387 Baseline Moving sic@
                                                  template (omit) ServingPLMNRateControl p_ServingPLMNRateCtrl := omit,
                                                  template (omit) ATSSSContainer p_ATSSSContainer := omit,
                                                  template (omit) ControlPlaneInd p_ControlPlaneInd := omit,
                                                  template (omit) IPHeaderCompressionConfig p_IPHeaderCompConfig := omit,
                                                  template (omit) EthernetHeaderCompressConfig p_EthernetHeaderCompConfig := omit) return template (value) NG_NAS_DL_Pdu_Type
  {
    var template (value) NG_NAS_DL_Message_Type v_GSMMsg;
    var template (value) PDU_SessionType v_PDU_SessionType;
    var B3_Type v_PDN_TypeToBeUsed;
    var PDN_AddressInfo_Type v_PDN_Address;
    var template (omit) PDU_Address v_PDU_Address := omit;
    var template (omit) AlwaysOnPDUSessionInd v_AlwaysOn := omit;
    var template (value) QoS_Rules v_QoS_Rules := f_BuildDefaultQoSRules ("1"); // @sic R5-202585 sic@
    var template (value) QoSFlowDescr v_QoSFlowDescr;
    var template (omit) ExtdProtocolConfigOptions v_ExtdPCO;
    var template (omit) MappedEPSBearerContexts v_MappedEPSBearerContexts := omit;
    var template (value) MappedEPSContext v_MappedEPSContext;
    var bitstring v_EPSParam; // If used
    var O1_Type v_EPSBearerId := '00'O; // If used
    var boolean v_InterworkWithEPS := false;
    var template (value) QoS_Flow v_QoS_Flow;
    
    if (isvalue(p_PDU_SessionType)) {
      v_PDU_SessionType := valueof(p_PDU_SessionType);
    } else {
      v_PDU_SessionType := cs_PDU_SessionType (p_GSM_MobilityInfo.SessionType);
    }
        
    if (isvalue(p_GMM_MobilityInfo.GMMCap) and p_InterworkWithEPS) { // @sic R5s190464 R5s191172, R5s201367 sic@
      if (p_GMM_MobilityInfo.GMMCap.s1Cap == '1'B) { // @sic R5-195328 sic@
        v_InterworkWithEPS := true;
        if (isvalue(p_GSM_MobilityInfo.EPS_Bearer)) { // @sic R5-204400 sic@
          v_EPSBearerId := hex2oct(int2hex(valueof(p_GSM_MobilityInfo.EPS_Bearer), 1) & '0'H);
        } else {
          FatalError(__FILE__, __LINE__, "EPS Supported but no bearer id calculated");
        }
        if (p_GSM_MobilityInfo.PDUType == IMS_DNN) {    // @sic R5-205939 sic@
          v_EPSParam := encvalue(f_Get_508_EPSQoS(DEF_2));
          v_MappedEPSContext := cs_OneMappedEPSContext(v_EPSBearerId, cs_EPSParameter_QoS(bit2oct(v_EPSParam)));
        } else {
          v_EPSParam := encvalue(f_Get_508_EPSQoS(DEF_1));
          v_MappedEPSContext := cs_OneMappedEPSContext(v_EPSBearerId, cs_EPSParameter_QoS(bit2oct(v_EPSParam)));
        }
        v_MappedEPSBearerContexts := cs_OneMappedEPSBearerContextList (v_MappedEPSContext);
      }
    }
    
....



Change 2:
	Function name
	EPSParameters

	Reason for change
	Only one length field should be included in the MappedEPSBearerContext when the parameter is a mapped EPS QoS.

	Summary of change
	Allow the EPSParameters length field to be omitted, so that the length of the parameter can be calculated and included in the octetstring

	TTCN module
	NG_NASTypeDefs



	  type record EPSParameters {
    O1_Type                    id,
    Type4Length_Type           lenOfContent  optional,
    octetstring                epsParam
  };



Change 3:
	Function name
	cs_EPSParameter_QoS

	Reason for change
	Only one length field should be included in the MappedEPSBearerContext when the parameter is a mapped EPS QoS.

	Summary of change
	New template as required in change 1.

	TTCN module
	NG_NASTemplates



	  template (value) EPSParameters cs_EPSParameter_QoS (octetstring p_Param) :=
  { 
    id              := '01'O,
    lenOfContent    := omit, // this field is included in the EPS QoS template
    epsParam        := p_Param
  };




